
1. [bookmark: _Toc122434485][bookmark: _Toc110869828]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 7.21 which is part of the IMS NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Shaun Harry
	shaun.harry@keysight.com
	Verification Test Summary 
Test Case: 	7.21
ATS Version:	iwd-TTCN3-B2020-09_D12wk23
System Simulator used:	Keysight S8704A Protocol Conformance Toolset
UE used:	Qualcomm SM 8450+SDX65
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc110869829]Corrections required
1.1 [bookmark: _Toc108637068][bookmark: _Toc110869830]Change 1
	Function name
	f_TC_7_21_NR5GC_IMS1

	Reason for change
	1. Currently when step 1 is initiated to add video to the ongoing call, the PDU session modification to establish QoS for the audio call may still be ongoing in IPCAN.
2. Currently at step 1 when the SS sends re-INVITE to add video to the ongoing speech call, the SDP offer and subsequent answer are derived from the sequence A.16.2 (MTSI MT video call). SDP contents from this sequence are correct for the newly added video stream only – the already established audio stream already has its own set of parameters (payload types, bandwitch, fmtp etc) which were established in an MO direction and should be maintained in the MT re-INVITE.

	Summary of change
	1. Add short delay before starting step 1 to let IPCAN procedures complete.
Note : This is added because for MO call there is no trigger back to the test case when resource reservation steps have completed. This synchornisation is not needed when preconditions are used because in this case UE will not send UPDATE until resource reservation is completed, providing the necessary snchronisation. This delay is added to avoid IPCAN needing to know if preconditions are in use.

2. Pass SDP from the initial INVITE into function  f_IMS_BuildSDP_MTCallAudioVideo_RX(), see change 2.


	TTCN module
	IMS_NR5GC_CallControlTestcases.ttcn

	MCC160 Comment
	



Before Change:
	  function f_TC_7_21_NR5GC_IMS1() runs on IMS_PTC
  { // MTSI MO Voice Call / add video and remove video / without preconditions at both originating UE and terminating UE / 5GS
    var template (omit) IMS_DATA_RSP v_IMS_DATA_RSP;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdpAudio;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdpVideo;
    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;
    var template (value) SDP_Message v_SdpOffer;
    var template (present) SDP_Message v_ExpectedSdpAnswer;
    var template (present) IMS_Response_Type v_Response183;
    var IMS_Response_Type v_Response;
    var SDP_Message v_SdpAnswer;
    var charstring v_CallIdentification;

    timer t_Timer := 2.0;

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);
    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall, -, normalCall_ReturnId);

    v_InviteRequestWithSdpAudio := f_IMS_MOCallSetup_NR5GC_A_4_2_Step1();
    f_IMS_MOCallSetup_NR5GC_A_4_2_Step2_8(v_InviteRequestWithSdpAudio);
    v_CallIdentification := f_UT_ReceivedIMSCallId(MMI);

    f_IMS_TestBody_Set(true);

    v_SdpOffer := f_IMS_BuildSDP_InitialOfferAddVideo_TX(EVS_16000, tsc_IMS_SDP_VideoCodecList_H265_only, v_InviteRequestWithSdpAudio.SdpOffer, tsc_SDP_WithoutPreconditions);
    v_ExpectedSdpAnswer := f_IMS_BuildSDP_MTCallAudioVideo_RX(AddVideo, InitialAnswer, EVS_16000, H265_90000, tsc_SDP_WithoutPreconditions);

    // @siclog "Step 1" siclog@                 INVITE
    v_InviteRequestWithSdpVideo := f_IMS_MTCallSetup_Video_SendInvite(AddVideo, v_SdpOffer);

    
………………
   
  }




After Change:
	  function f_TC_7_21_NR5GC_IMS1() runs on IMS_PTC
  { // MTSI MO Voice Call / add video and remove video / without preconditions at both originating UE and terminating UE / 5GS
    var template (omit) IMS_DATA_RSP v_IMS_DATA_RSP;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdpAudio;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdpVideo;
    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;
    var template (value) SDP_Message v_SdpOffer;
    var template (present) SDP_Message v_ExpectedSdpAnswer;
    var template (present) IMS_Response_Type v_Response183;
    var IMS_Response_Type v_Response;
    var SDP_Message v_SdpAnswer;
    var charstring v_CallIdentification;

    timer t_Timer := 2.0;

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);
    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall, -, normalCall_ReturnId);

    v_InviteRequestWithSdpAudio := f_IMS_MOCallSetup_NR5GC_A_4_2_Step1();
    f_IMS_MOCallSetup_NR5GC_A_4_2_Step2_8(v_InviteRequestWithSdpAudio);
    v_CallIdentification := f_UT_ReceivedIMSCallId(MMI);

    f_IMS_TestBody_Set(true);

    v_SdpOffer := f_IMS_BuildSDP_InitialOfferAddVideo_TX(EVS_16000, tsc_IMS_SDP_VideoCodecList_H265_only, v_InviteRequestWithSdpAudio.SdpOffer, tsc_SDP_WithoutPreconditions);
    v_ExpectedSdpAnswer := f_IMS_BuildSDP_MTCallAudioVideo_RX(AddVideo, InitialAnswer, EVS_16000, H265_90000, tsc_SDP_WithoutPreconditions, v_InviteRequestWithSdpAudio.SdpOffer);

    f_Delay(2.0);

    // @siclog "Step 1" siclog@                 INVITE
    v_InviteRequestWithSdpVideo := f_IMS_MTCallSetup_Video_SendInvite(AddVideo, v_SdpOffer);

    
…………
   
  }  



1.2 [bookmark: _Toc108637069][bookmark: _Toc110869831]Change 2
	Function name
	f_IMS_BuildSDP_MTCallAudioVideo_RX

	Reason for change
	Currently at step 1 when the SS sends re-INVITE to add video to the ongoing speech call, the SDP offer and subsequent answer are derived from the sequence A.16.2 (MTSI MT video call). SDP contents from this sequence are correct for the newly added video stream only – the already established audio stream already has its own set of parameters (payload types, bandwitch, fmtp etc) which were established in an MO direction and should be maintained in the MT re-INVITE.

	Summary of change
	Changed the function f_IMS_BuildSDP_MTCallAudioVideo_RX() to accept an extra parameter for the SDP from the initial INVITE, and use this to select the audio payload type number and fmtp parameters.

Note : This information is already used to create the updated SDP offer in f_SDP_MediaDescriptionSubsequentOfferAudioTX() (called from  f_IMS_BuildSDP_InitialOfferAddVideo_TX()), but also needs to be considered when creating the expected answer.

Note : An associated prose CR on 34.229-5 will be raised at RAN5#96.

	TTCN module
	IMS_SDP_MessagesVideo.ttcn

	MCC160 Comment
	



Before Change:
	  function f_IMS_BuildSDP_MTCallAudioVideo_RX(VideoCallTypeMT_Type p_VideoCallType,
                                              template (omit) SDP_KindOfAnswer_Type p_KindOfAnswer := omit,
                                              IMS_SDP_AudioCodec_Type p_AudioCodec := AMR_16000,
                                              IMS_SDP_VideoCodec_Type p_VideoCodec := H264_90000,
                                              boolean p_WithPreconditions := true)
    runs on IMS_PTC
    return template (present) SDP_Message
  { /* @sic R5-172930: function restructured to cope with different flavours ('normal' IMS, WLAN; Setup video call, add video, remove video; INVITE, UPDATE) sic@ */
    /* @sic R5-172930: p_FmtAudio, p_FmtVideo removed sic@ */
    /* @sic R5-213524: p_AudioCodec, p_VideoCodec, p_WithPreconditions sic@ */
    var template (present) SDP_attribute_rtpmap_codec v_AudioCodec;
    var template (present) SDP_attribute_rtpmap_codec v_VideoCodec;
    var template (present) SDP_attribute_list v_MediaAttributes_Audio;
    var template (present) SDP_attribute_list v_MediaAttributes_Video;
    var template (present) SDP_attribute_list v_PreconditionAttributes_Audio := {}; // @sic R5-170782 sic@
    var template (present) SDP_attribute_list v_PreconditionAttributes_Video := {}; // @sic R5-170782 sic@
    var template (present) SDP_attribute_list v_PreconditionAttributes_Final := cr_SDP_PreconditionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_mandatory);  /* preconditions when resources are fully available */
    var template (present) SDP_attribute_fmtp_format_specific_params v_FmtpParams_Audio := ?;
    var template (present) SDP_attribute_fmtp_format_specific_params v_FmtpParams_Video := ?;
    var charstring v_FmtAudio;
    var charstring v_FmtVideo := tsc_SDP_FmtVideo;               /* @sic R5-197050: tsc_SDP_FmtVideo instead of fl_IMS_BuildSDP_FmtVideo sic@ */
    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();
    var boolean v_IsInitialAnswer := match(InitialAnswer, p_KindOfAnswer);
   


    v_AudioCodec := f_SDP_AttributeRtpmapAudioRX(p_AudioCodec);  // @sic R5-168058; R5-168142 sic@
    v_FmtAudio := f_SDP_AudioCodec_GetFmt(p_AudioCodec);         /* @sic R5-197050: fix mapping of fmt values sic@ */
    
    
    if (p_WithPreconditions) {
      if (not v_IsWLAN and v_IsInitialAnswer) {  // i.e. 183 Session Progress (C.26 step 4 or 17.2 step 3)
        v_PreconditionAttributes_Video := cr_SDP_PreconditionAttributesWithConf((c_none, c_sendrecv), c_none, c_mandatory, c_mandatory);    // @sic R5-216443: NR5GC sic@
        
        if (p_VideoCallType == SetupVideoCall) {                 // C.26 step 4: same preconditions for video and audio
          v_PreconditionAttributes_Audio := v_PreconditionAttributes_Video;
        } else {                                                 // 17.2 step 3: audio is there already
          v_PreconditionAttributes_Audio := v_PreconditionAttributes_Final;
        }
      } else {  // WLAN or 200 OK
        v_PreconditionAttributes_Audio := v_PreconditionAttributes_Final;
        v_PreconditionAttributes_Video := v_PreconditionAttributes_Final;
      }
    }
    
    if ((p_AudioCodec == EVS_16000) and v_IsInitialAnswer) {   /* @sic R5-213524 sic@ */
      v_FmtpParams_Audio := cr_Fmtp_ParamList(cr_EVS_Fmtp_AudioParamsBrBwModeSetMaxRed("13.2", "swb")); // @sic R5-221471: cr_EVS_Fmtp_AudioParamsBrBwModeSetMaxRed instead of cr_EVS_Fmtp_AudioParamsBrBwMaxRed sic@
    }
    select (p_VideoCodec) {                                    /* @sic R5-213524 sic@ */
      case (H264_90000) {
        v_VideoCodec := cr_RTPMAP_H264_90000;
        v_FmtpParams_Video := cr_SDP_FmtpParamsH264_MT;
      }
      case (H265_90000) {
        v_VideoCodec := cr_RTPMAP_H265_90000;    // @sic R5s220235 sic@
        v_FmtpParams_Video := cr_SDP_FmtpParamsH265("1", "93");
      }
    }

    v_MediaAttributes_Audio := {
      cr_SDP_Attribute_rtpmap(v_FmtAudio, v_AudioCodec),
      cr_SDP_Attribute_fmtp(v_FmtAudio, v_FmtpParams_Audio)
    };
    v_MediaAttributes_Video := {
      cr_SDP_Attribute_rtpmap(v_FmtVideo, v_VideoCodec),
      cr_SDP_Attribute_fmtp(v_FmtVideo, v_FmtpParams_Video)  /* @sic R5s141329 - MCC160 implementation sic@
                                                                @sic R5-150703: same parameters for C.26 and 17.2 sic@ */
    };
    if (f_IMS_PTC_RanTypeIsNR5GC()) {    // @sic R5-216215 sic@
      v_MediaAttributes_Video := f_SDP_Attributes_Add_RX(v_MediaAttributes_Video, cr_SDP_Attribute_acfg(1, "t=1"));
    }

    return f_IMS_BuildSDP_AudioVideo_RX(-,
                                        cr_SDP_Media_Audio(v_FmtAudio),
                                        v_MediaAttributes_Audio,
                                        v_PreconditionAttributes_Audio,
                                        cr_SDP_Media_Video(v_FmtVideo),
                                        v_MediaAttributes_Video,
                                        v_PreconditionAttributes_Video);
  }




After Change:
	…
…
  import from IMS_CommonTemplates all;
…
…


function f_IMS_BuildSDP_MTCallAudioVideo_RX(VideoCallTypeMT_Type p_VideoCallType,
                                              template (omit) SDP_KindOfAnswer_Type p_KindOfAnswer := omit,
                                              IMS_SDP_AudioCodec_Type p_AudioCodec := AMR_16000,
                                              IMS_SDP_VideoCodec_Type p_VideoCodec := H264_90000,
                                              boolean p_WithPreconditions := true,     
                                              template (omit) SDP_Message p_PrevSdpFromUE := omit)
    runs on IMS_PTC
    return template (present) SDP_Message
  { /* @sic R5-172930: function restructured to cope with different flavours ('normal' IMS, WLAN; Setup video call, add video, remove video; INVITE, UPDATE) sic@ */
    /* @sic R5-172930: p_FmtAudio, p_FmtVideo removed sic@ */
    /* @sic R5-213524: p_AudioCodec, p_VideoCodec, p_WithPreconditions sic@ */
    var template (present) SDP_attribute_rtpmap_codec v_AudioCodec;
    var template (present) SDP_attribute_rtpmap_codec v_VideoCodec;
    var template (present) SDP_attribute_list v_MediaAttributes_Audio;
    var template (present) SDP_attribute_list v_MediaAttributes_Video;
    var template (present) SDP_attribute_list v_PreconditionAttributes_Audio := {}; // @sic R5-170782 sic@
    var template (present) SDP_attribute_list v_PreconditionAttributes_Video := {}; // @sic R5-170782 sic@
    var template (present) SDP_attribute_list v_PreconditionAttributes_Final := cr_SDP_PreconditionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_mandatory);  /* preconditions when resources are fully available */
    var template (present) SDP_attribute_fmtp_format_specific_params v_FmtpParams_Audio := ?;
    var template (present) SDP_attribute_fmtp_format_specific_params v_FmtpParams_Video := ?;
    var charstring v_FmtAudio;
    var charstring v_FmtVideo := tsc_SDP_FmtVideo;               /* @sic R5-197050: tsc_SDP_FmtVideo instead of fl_IMS_BuildSDP_FmtVideo sic@ */
    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();
    var boolean v_IsInitialAnswer := match(InitialAnswer, p_KindOfAnswer);
    var SDP_Message v_PrevSdpFromUE;
    var SDP_media_desc v_MediaDesc;
    var template (value) SemicolonParam_List v_EVS_FmtpParams;
    var template (omit) SDP_attribute v_FmtpAttributeEVS;
    var charstring v_ReceivedBrVal ;  /* @sic R5-195318: "br" instead of "br-recv", "br-send" sic@ */
    var charstring v_ReceivedBwVal;
    


    v_AudioCodec := f_SDP_AttributeRtpmapAudioRX(p_AudioCodec);  // @sic R5-168058; R5-168142 sic@
    v_FmtAudio := f_SDP_AudioCodec_GetFmt(p_AudioCodec);         /* @sic R5-197050: fix mapping of fmt values sic@ */
    
    if (isvalue(p_PrevSdpFromUE)) {
      v_PrevSdpFromUE := valueof(p_PrevSdpFromUE);
      v_MediaDesc := valueof(f_IMS_SDP_MediaList_GetMediaStream(v_PrevSdpFromUE.media_list, c_audio));
      v_FmtAudio := f_SDP_MediaDescr_GetAudioCodec(v_MediaDesc, p_AudioCodec);
    }//ADDED
    
    
    if (p_WithPreconditions) {
      if (not v_IsWLAN and v_IsInitialAnswer) {  // i.e. 183 Session Progress (C.26 step 4 or 17.2 step 3)
        v_PreconditionAttributes_Video := cr_SDP_PreconditionAttributesWithConf((c_none, c_sendrecv), c_none, c_mandatory, c_mandatory);    // @sic R5-216443: NR5GC sic@
        
        if (p_VideoCallType == SetupVideoCall) {                 // C.26 step 4: same preconditions for video and audio
          v_PreconditionAttributes_Audio := v_PreconditionAttributes_Video;
        } else {                                                 // 17.2 step 3: audio is there already
          v_PreconditionAttributes_Audio := v_PreconditionAttributes_Final;
        }
      } else {  // WLAN or 200 OK
        v_PreconditionAttributes_Audio := v_PreconditionAttributes_Final;
        v_PreconditionAttributes_Video := v_PreconditionAttributes_Final;
      }
    }
    
    if ((p_AudioCodec == EVS_16000) and v_IsInitialAnswer) {   /* @sic R5-213524 sic@ */
      v_FmtpParams_Audio := cr_Fmtp_ParamList(cr_EVS_Fmtp_AudioParamsBrBwModeSetMaxRed("13.2", "swb")); // @sic R5-221471: cr_EVS_Fmtp_AudioParamsBrBwModeSetMaxRed instead of cr_EVS_Fmtp_AudioParamsBrBwMaxRed sic@
    }
    select (p_VideoCodec) {                                    /* @sic R5-213524 sic@ */
      case (H264_90000) {
        v_VideoCodec := cr_RTPMAP_H264_90000;
        v_FmtpParams_Video := cr_SDP_FmtpParamsH264_MT;
      }
      case (H265_90000) {
        v_VideoCodec := cr_RTPMAP_H265_90000;    // @sic R5s220235 sic@
        v_FmtpParams_Video := cr_SDP_FmtpParamsH265("1", "93");
      }
    }

    if (isvalue(p_PrevSdpFromUE)) {
      v_FmtpAttributeEVS := f_SDP_AttributeList_GetAttribute(v_MediaDesc.attributes, cr_SDP_Attribute_fmtp(v_FmtAudio));
      v_ReceivedBrVal := f_SIP_SemicolonParamList_GetParamValue(v_FmtpAttributeEVS.fmtp.params.paramList, "br");  /* @sic R5-195318: "br" instead of "br-recv", "br-send" sic@ */
      v_ReceivedBwVal := f_SIP_SemicolonParamList_GetParamValue(v_FmtpAttributeEVS.fmtp.params.paramList, "bw"); 
      v_EVS_FmtpParams := cr_EVS_Fmtp_AudioParamsBrBw(v_ReceivedBrVal, v_ReceivedBwVal);    /* @sic R5-195318 sic@ @sic R5-210056: NR5GC sic@ */

      v_MediaAttributes_Audio:= {
        cr_SDP_Attribute_rtpmap(v_FmtAudio, cr_RTPMAP_EVS_16000),
        cr_SDP_Attribute_fmtp(v_FmtAudio, cs_Fmtp_ParamList(v_EVS_FmtpParams))
      };
    }
    else
    {

      v_MediaAttributes_Audio := {
        cr_SDP_Attribute_rtpmap(v_FmtAudio, v_AudioCodec),
        cr_SDP_Attribute_fmtp(v_FmtAudio, v_FmtpParams_Audio)
      };
    }//ADDED
    v_MediaAttributes_Video := {
      cr_SDP_Attribute_rtpmap(v_FmtVideo, v_VideoCodec),
      cr_SDP_Attribute_fmtp(v_FmtVideo, v_FmtpParams_Video)  /* @sic R5s141329 - MCC160 implementation sic@
                                                                @sic R5-150703: same parameters for C.26 and 17.2 sic@ */
    };
    if (f_IMS_PTC_RanTypeIsNR5GC()) {    // @sic R5-216215 sic@
      v_MediaAttributes_Video := f_SDP_Attributes_Add_RX(v_MediaAttributes_Video, cr_SDP_Attribute_acfg(1, "t=1"));
    }

    return f_IMS_BuildSDP_AudioVideo_RX(-,
                                        cr_SDP_Media_Audio(v_FmtAudio),
                                        v_MediaAttributes_Audio,
                                        v_PreconditionAttributes_Audio,
                                        cr_SDP_Media_Video(v_FmtVideo),
                                        v_MediaAttributes_Video,
                                        v_PreconditionAttributes_Video);
  }  






1. [bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc110869832]Branches executed
[bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]The test case was executed on NR band n41 using NR ciphering nea2 and integrity algorithm nia2
3 [bookmark: _Toc110869833]Execution Log Files
3.1 [bookmark: _Toc110869834]Qualcomm SM 8450+SDX65
The Qualcomm SM 8450+SDX65  UE passed this test case on Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
Keysight/TC_7_21_LOG.html
· PICS/PIXIT parameter file:
Keysight/TC_7_21_PIXIT.xml



4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc110869835]References 
	[1]
	R5s220940 : Supporting information for addition of IMS NR5GC call control test case 7.21



